

NAG Fortran Library Routine Document

G05QAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05QAF generates a random orthogonal matrix.

2 Specification

```
SUBROUTINE G05QAF(SIDE, INIT, M, N, A, LDA, IGEN, ISEED, WK, IFAIL)
INTEGER          M, N, LDA, IGEN, ISEED(4), IFAIL
real            A(LDA,N), WK(*)
CHARACTER*1      SIDE, INIT
```

3 Description

G05QAF pre- or post-multiplies an m by n matrix A by a random orthogonal matrix U , overwriting A . The matrix A may optionally be initialised to the identity matrix before multiplying by U , hence U is returned. U is generated using the method of Stewart (1980). The algorithm can be summarized as follows.

Let $x_1, x_2, \dots, x_{n-1}$ follow independent multinormal distributions with zero mean and variance $I\sigma^2$ and dimensions $n, n-1, \dots, 2$; let $H_j = \text{diag}(I_{j-1}, H_j^*)$, where I_{j-1} is the identity matrix and H_j^* is the Householder transformation that reduces x_j to $r_{jj}e_1$, e_1 being the vector with first element one and the remaining elements zero and r_{jj} being a scalar, and let $D = \text{diag}(\text{sign}(r_{11}), \text{sign}(r_{22}), \dots, \text{sign}(r_{nn}))$. Then the product $U = DH_1H_2 \dots H_{n-1}$ is a random orthogonal matrix distributed according to the Haar measure over the set of orthogonal matrices of n . See Stewart (1980), Theorem 3.3.

One of the initialisation routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05QAF.

4 References

Stewart G W (1980) The efficient generation of random orthogonal matrices with an application to condition estimates *SIAM J. Numer. Anal.* **17** 403–409

5 Parameters

1: SIDE – CHARACTER*1 *Input*

On entry: indicates whether the matrix A is multiplied on the left or right by the random orthogonal matrix U .

If SIDE = 'L', the matrix A is multiplied on the left, i.e., pre-multiplied.

If SIDE = 'R', the matrix A is multiplied on the right, i.e., post-multiplied.

Constraint: SIDE = 'L' or 'R'.

- 2: INIT – CHARACTER*1 *Input*
On entry: indicates whether or not A should be initialised to the identity matrix.
 If INIT = 'I', A is initialised to the identity matrix.
 If INIT = 'N', A is not initialised and the matrix A must be supplied in A.
Constraint: INIT = 'I' or 'N'.
- 3: M – INTEGER *Input*
On entry: the number of rows of the matrix A, m .
Constraint: if SIDE = 'L', $M > 1$ else $M \geq 1$.
- 4: N – INTEGER *Input*
On entry: the number of columns of the matrix A, n .
Constraint: if SIDE = 'R', $N > 1$ else $N \geq 1$.
- 5: A(LDA,N) – *real* array *Input/Output*
On entry: if INIT = 'N', A must contain the matrix A.
On exit: the matrix UA when SIDE = 'L' or the matrix AU when SIDE = 'R'.
- 6: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which G05QAF is called.
Constraint: $LDA \geq M$.
- 7: IGEN – INTEGER *Input*
On entry: must contain the identification number for the generator to be used to return a pseudo-random number and should remain unchanged following initialisation by a prior call to one of the routines G05KBF or G05KCF.
- 8: ISEED(4) – INTEGER array *Input/Output*
On entry: contains values which define the current state of the selected generator.
On exit: contains updated values defining the new state of the selected generator.
- 9: WK(*) – *real* array *Workspace*
Note: the dimension of the array WK must be at least $2 \times M$ if SIDE = 'L' or $2 \times N$ if SIDE = 'R'.
- 10: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $M < 1$,
or $N < 1$,
or $LDA < M$.

IFAIL = 2

On entry, $SIDE \neq 'L'$ or $'R'$,
or $INIT \neq 'I'$ or $'N'$.

IFAIL = 3

On entry, an orthogonal matrix of dimension 1 has been requested.

7 Accuracy

The maximum error in $U^T U$ should be a modest multiple of *machine precision* (see Chapter X02).

8 Further Comments

G05QBF computes a random correlation matrix from a random orthogonal matrix.

9 Example

Following initialisation of the pseudo random number generator by a call to G05KBF, a 4 by 4 orthogonal matrix is generated using the $INIT = 'I'$ option and the result printed.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G05QAF Example Program Text
*      Mark 20 Release. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER          N, M, LDA
      PARAMETER        (N=4,M=4,LDA=10)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, IGEN, J
*      .. Local Arrays ..
      real             A(LDA,N), WK(2*N)
      INTEGER          ISEED(4)
*      .. External Subroutines ..
      EXTERNAL         G05KBF, G05QAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05QAF Example Program Results'
      WRITE (NOUT,*)

*
*      IGEN identifies the stream.
      IGEN = 1
*      Initialise the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 423446
      ISEED(4) = 742355
      CALL G05KBF(IGEN,ISEED)
*
*      IFAIL = 0
*
```

```
      CALL G05QAF('Right','Initialize',M,N,A,LDA,IGEN,ISEED,WK,IFAIL)
*
      DO 20 I = 1, M
        WRITE (NOUT,99999) (A(I,J),J=1,N)
20    CONTINUE
      STOP
*
99999 FORMAT (1X,4F9.3)
      END
```

9.2 Program Data

None.

9.3 Program Results

G05QAF Example Program Results

-0.219	-0.197	-0.413	-0.862
0.438	0.006	0.762	-0.478
0.699	-0.627	-0.322	0.120
-0.521	-0.754	0.381	0.122
